





DIFF_STAGE_Thd_VS_VIN

HOW TO CALCULATED AND PLOT HARMONIC INTERCEPTS.



```

=====Run Transient=====
THD% versus VIN_vpk
=====Create Unknown_display=====
=====Create VinVals_Array=====
=====Create Thd Vals_Array=====
=====use unknown to store data=====
k = 0 unknown.Vin = 0.001 thd_percent= 0.00310418
=====use unknown to store data=====
k = 1 unknown.Vin = 0.01 thd_percent= 0.308224
=====use unknown to store data=====
k = 2 unknown.Vin = 0.03 thd_percent= 2.58693
=====use unknown to store data=====
k = 3 unknown.Vin = 0.05 thd_percent= 6.34364
=====use unknown to store data=====
k = 4 unknown.Vin = 0.08 thd_percent= 12.7367
=====use unknown to store data=====
k = 5 unknown.Vin = 0.1 thd_percent= 16.6809
=====use unknown to store data=====
k = 6 unknown.Vin = 0.11 thd_percent= 18.4652
=====use unknown to store data=====
k = 7 unknown.Vin = 0.3 thd_percent= 33.7664
=====Plot_Outputs=====
=====Calculate Slopes=====


```

Anonymous
unknown Thu Feb 17 21:21:27 2011

Index	VinVals	unknown.fund	unknown.third
0	1.000000e-03	1.931907e-06	5.597175e-11
1	1.000000e-02	1.914262e-05	5.900151e-08
2	3.000000e-02	5.357803e-05	1.385342e-06
3	5.000000e-02	7.933356e-05	5.017158e-06
4	8.000000e-02	1.023417e-04	1.286460e-05
5	1.000000e-01	1.106689e-04	1.803445e-05
6	1.100000e-01	1.135374e-04	2.036444e-05
7	3.000000e-01	1.256616e-04	3.796293e-05

=====Create slope Array=====
intercept = 1.733480e-01
Done.

DIFF_STAGE_Thd_VS_VIN



```
.OPTIONS GMIN=1e-18 METHOD=trap srcsteps = 1 gminsteps = 1
*****
VCC      VC      0      DC      10
VTime    VTime   0      DC      0      PWL( 0 0 1 1)
Vfreq1   Vfreq1  0      DC      1000
Vmag     Vmag    0      DC      .1
BTRI     TRI     0      V      = 2*acos(cos(6.2831*V(VFreq1)*V(VTime)))/3.141592653589793 -1
BVIN     VIN     0      V      = V(Vmag)*cos(6.2831*V(VFreq1)*V(VTime))
I1       VEN1   0      100u
QN1      VCN1   VIN     VEN1   NPN1   1.00
QN2      VCN2   0      VEN1   NPN1   1.00
R1       VCN1   VC      1
R2       VCN2   VC      1
BOUT     OUT    0      V      = V(VCN1)-V(VCN2)
```

```
.control
set pensize = 2
echo "=====Run_Transient=====
tran 25u 5m 0 1u
plot out
echo "THD% versus VIN_vpk"
echo "=====Create_Unknown_display=====
setplot new
echo "=====Create_VinVals_Array=====
compose VinVals values 1m 10m 30m 50m 80m 100m 110m 300m
settype voltage VinVals
let NoOfVin = length(VinVals)
compose kk values 1 2 3
echo "=====Create_Thd_Vals_Array=====
let tthd = vector($NoOfVin)*0
let fund = vector($NoOfVin)*0
let third = vector($NoOfVin)*0
echo "=====Begin_1_loops=====
let k = 1
while (k <= NoOfVin )
let unknown.kk[0] = k
let Vin = VinVals[k-1]
let thdsq = 0
*echo "Vin = $&Vin"
alter Vmag dc = $&Vin
tran 25u 5m 0 1u
linearize
set specwindow= "rectangular"
spec 200 8k 200 v(out)
let thdsq =mag(out[9])^2 +mag(out[14])^2 +mag(out[19])^2 +mag(out[24])^2
let thd_percent= 100*sqrt(thdsq)/mag(out[4])
echo "=====use_unknown_to_store_data=====
let k2 = unknown.kk[0]-1
echo "k = $&k2 unknown.Vin = $&unknown.Vin thd_percent= $&thd_percent"
let k2 = unknown.kk[0] -1
let unknown.tthd[$&k2] = thd_percent
let unknown.fund[$&k2] = mag(out[4])
let unknown.third[$&k2] = mag(out[14])
destroy
destroy
destroy
let k = k + 1
endwhile
echo "=====Plot_Outputs=====
setscale VinVals
plot unknown.tthd loglog title "THD_% vs Vin_pK and Temp_C"
plot unknown.fund unknown.third loglog title "THD_% vs Vin_pK and Temp_C"
echo "=====Calculate_Slopes=====
print VinVals unknown.fund unknown.third
let fpow = ln(fund[1]/fund[0])/ln(VinVals[1]/VinVals[0])
let thrpow = ln(third[1]/third[0])/ln(VinVals[1]/VinVals[0])
let fundgain = fund[0]/exp(fpow*ln(VinVals[0]))
let thrgain = third[0]/exp(thrpow*ln(VinVals[0]))
*print thrgain
echo "=====Create_slope_Array=====
let thirdp = vector($NoOfVin)*0
let fundp = vector($NoOfVin)*0
let k = 0
while (k < NoOfVin )
let thirdp[$&k] = thrgain*exp(thrpow*ln(VinVals[$&k]))
let fundp[$&k] = fundgain*exp(fpow*ln(VinVals[$&k]))
let k = k + 1
endwhile
plot fund third thirdp fundp loglog title "THD_% vs Vin_pK and Temp_C"
let intercept = exp(ln(thrgain/fundgain)/(fpow-thrpow))
print intercept
.endc
```

```
*=====
.model NPN1 NPN( BF=2100 VAF=216 )
.model PNP1 PNP( BF=2100 VAF=21 )
.end
```